
// Commentaires : Envoi d'une donnée ou exécution d'une
// commande (deux boutons sur une page web)
// en réponse à une requête (REST)
// arduino/analog/val1 : lecture de l'entrée analogique 0
// arduino/analog/val2 : lecture de l'entrée analogique 1
// arduino/analog/val3 : lecture de l'entrée analogique 2
// arduino/analog/val4 : lecture de l'entrée analogique 3
// arduino/digital/13/x : x = 0 ou 1 appliqué à la pin 13
//
// Matériel : Carte Arduino Uno Wifi
// Version: 1.1
// Date: 15/3/2019
// Fichier : testUnoWifi4.ino
// ---------------------------------------------------
// Bibliothèques
#include <Wire.h>
#include <UnoWiFiDevEd.h>
// ---------------------------------------------------
// Variables globales
int val1 = 0, val2 = 0, val3 = 0, val4 = 0;
// ---------------------------------------------------
void setup()
{
  Wifi.begin(); // Démarrage de la carte Wifi
}
// ---------------------------------------------------
void loop()
{
  while (Wifi.available()) // Si on reçoit une requête contenant : 
  {			// Ex1 lecture A0: arduino/analog/val1 => Wifi contient 'analog/val1' 
    process(Wifi);	// Ex2 écriture '1' sur D11 : arduino/digital/11/x => Wifi contient 'digital/11/x'	
  }			// (x= 0 ou 1) alors on appelle process et on lui passe Wifi
  delay(50);
}
// ---------------------------------------------------
void process(WifiData client)
{
  // On lit le contenu de Wifi jusqu'au premier '/' et on le place dans command
  String command = client.readStringUntil('/');

  // si command = analog (Ex1) alors on appelle la fonction analogCommand et on lui passe client
  if (command == "analog"){
    analogCommand(client); // client = 'val1'
  }
  // si command = digital (Ex2) alors on appelle la fonction digitalCommand et on lui passe client
  if (command == "digital"){
    digitalCommand(client); // client = '11/x' (x= 0 ou 1)
  }
}
// ---------------------------------------------------



void analogCommand(WifiData client) { // Rappel (Ex1) client = 'val1'
  // On lit le contenu de client jusqu'à la fin et on le place dans command
  String command = client.readStringUntil('\r');
  // si command = val1
  if (command == "val1"){
    val1 = analogRead(0); // On lit l'entrée A0 et on place sa valeur dans val1
    client.println("HTTP/1.1 200 OK\n"); // On construit et on envoie la réponse à la requête
    client.print(val1); // contenant la réponse
    client.print(EOL); // terminateur
  }
  if (command == "val2"){
    val2 = analogRead(1);
    client.println("HTTP/1.1 200 OK\n");
    client.print(val2);
    client.print(EOL); 
  }
  if (command == "val3"){
    val3 = analogRead(2);
    client.println("HTTP/1.1 200 OK\n");
    client.print(val3);
    client.print(EOL);
  }
  if (command == "val4"){
     val4 = analogRead(3);
    client.println("HTTP/1.1 200 OK\n");
    client.print(val4);
    client.print(EOL);
  }
}
// ---------------------------------------------------
void digitalCommand(WifiData client) { // Rappel (Ex2) client = '11/x' x= 0 ou 1
  int pin, value;
  // Lecture du numéro de la broche
  pin = client.parseInt(); // pin = 11
  
  // Lecture du caractère suivant si c'est un '/' alors
  if (client.read() == '/'){
    value = client.parseInt(); // on lit la valeur de x (0 ou 1)
    digitalWrite(pin, value); // et on écrit x sur pin
  }
  else{
    value = digitalRead(pin); // sinon on lit la valeur de la broche
  }

  // On envoie une réponse au client
  client.println("Status: 200 OK\n");
  client.print(value);
  client.print(EOL); // terminateur
}

